UNITED STATES DEPARTMENT OF THE INTERIOR
GEOLOGICAL SURVEY

A Hewlett-Packard 9830A BASIC language program for Plumbotectonics

by

Robert E. Zartman and Bruce R. Doe

U.S. Geological Survey

Denver, Colorado 80225

Open-File Report 80-1088

1980

Three broad geologic environments, or reservoirs, having appreciable amounts of U, Th, and Pb and sufficient longevity, are required by the plumbotectonics model. These resevoirs, the mantle, upper continental crust, and lower continental crust, and an additional short-lived mixing orogene provide the necessary chemical environments to produce gross terrestrial isotopic patterns. The transfer of matter among reservoirs is treated as proceeding through cycles--called orogenies--of discrete, sequential operations that can be expressed by a series of BASIC program statements (paralleling the algebraic equations of the companion paper). From the sets of numerical data generated as matrix arrays, both the total mass distribution and the U, Th, and Pb isotopic contents among reservoirs can be determined as a function of time.

Two versions of the model are presented by Zartman and Doe (in press), which follow the same basic program form and differ only in certain parametric values. We list version I in its entirety in Appendix A, and supply the pertinent substitute lines for conversion to version II in Appendix B. In addition to yielding mass distribution and U, Th, and Pb isotopic contents of reservoirs, several output options are available for calculating isotopic ratios, mean age of the crust, and isotope concentrations of reservoirs. A separate subroutine is also provided, which modifies the main program, for computing the mean age of isotopes in the crust.

The program was written specifically for use with the Hewlett-Packard model 9830A minicomputer with 8K words of memory, but with a few modifications it could be adapted to several other comparable H-P models. If the appropriate tape cartridge is sent to us at Mailstop 963, Federal enter, Denver, Colorado, 80225, we can supply you with a copy of the program.

Table 1. Algebraic notations and their equivalent BASIC representation of the plumbotectonics model

Algebraic	BASIC	Description
$M^{m}_{\underline{t}}$, $M^{u(i)}_{\underline{t}}$, $M^{\ell(i)}_{\underline{t}}$	M[1,1], M [1,i] for i=2 to 12, M[1,i] for i=13 to 23	Mass of mantle, _ith_ segment of upper crust, and _ith_ segment of lower crust, respectively, immediately prior to an orogeny at time, t.
$M^{m}_{\overline{t}}$, $M^{u(i)}_{\overline{t}}$, $M^{\ell(i)}_{\overline{t}}$	M[4,1], M[4,i] for i=2 to 12, M[4,i] for i=13 to 23	Mass of mantle, _ith_ segment of upper crust, and _ith_ segment of lower crust, respectively, immediately following an orogeny at time, t.
M^{o}_{t}	MØ	Mass of orogene during an orogeny at time, t.
$\Delta M^{m}_{\underline{t}}$, $\Delta M^{u(i)}_{\underline{t}}$, $\Delta M^{\ell(i)}_{\underline{t}}$	M[2,1], M[2,i] for i=2 to 12, M[2,i] for i=13 to 23	Incremental mass of mantle, _ith_ segment of upper crust, and _ith_ segment of lower crust, respectively, used to create an orogene at time, t.
$\Delta M^{m}_{\overline{t}}$, $M^{u(j)}_{\overline{t}}$, $M^{\ell(j)}_{\overline{t}}$	M[3,1], M[4,j] (U) 2<j< 12, M[4,j] (L) 13<j<23	Incremental mass returned to the mantle, mass of a new (_jth_) segment of upper crust, and mass of a new (_jth_) segment of lower crust, respectively, representing the redistribution from an orogene at time, t.
$(M^{m}_{\underline{t}} / M^{m}_{\underline{t}})$	A[k] for k=1 to 11	Mass fractionation of mantle contributing to the creation of an orogene at time, t.

Table 1. Algebraic notations and their equivalent BASIC representation of the plumbotectonics model (continued)

Algebraic	BASIC	Description
f_v, f_h	B[k] for k=1 to 11, C[k] for k=1 to 11	Erosional (vertical) and areal (horizontal) functions, respectively. f_v is the fractional height of preexisting segments of <u>upper</u> crust and f_h is the fractional area of preexisting segments of <u>total</u> crust contributing to the creation of an orogene at time, t.
$^{\alpha}N^{m}_{\underline{t}}$, $^{\alpha}N^{u(i)}_{\underline{t}}$, $^{\alpha}N^{\ell(i)}_{\underline{t}}$	P[1,1], P[1,i] for i=2 to 12, P[1,i] for i=13 to 23	Moles of isotopic species α in the mantle, <u>ith</u> segment of lower crust, respectively, immediately prior to an orogeny at time, t.[1]
$^{\alpha}N^{m}_{\bar{t}}$, $^{\alpha}N^{u(i)}_{\bar{t}}$, $^{\alpha}N^{\ell(i)}_{\bar{t}}$	P[4,1], P[4,i] for i=2 to 12, P[4,i] for i=13 to 23	Moles of isotopic species α in the mantle, <u>ith</u> segment of lower crust, respectively, immediately following an orogeny at time, t.[1]
$^{\alpha}N^{o}_{t}$	PØ	Moles of isotopic species α in the orogene during an orogeny at time, t.[1]
$\Delta^{\alpha}N^{t}_{\underline{t}}$, $\Delta^{\alpha}N^{u(i)}_{\underline{t}}$, $\Delta^{\alpha}N^{\ell(i)}_{\underline{t}}$	P[2,1], P[2,i] for i=2 to 12, P[2,i] for i=13 to 23	Moles of isotopic species α contained in the incremental additions to the orogene from the mantle, <u>ith</u> segment of upper crust, and <u>ith</u> segment of lower crust, respectively, at time, t.[1]

Table 1. Algebraic notations and their equivalent BASIC representation of the plumbotectonics model (continued)

Algebraic	BASIC	Descriptions
${\Delta}^{\alpha}N^{m}_{\bar{t}}$, ${\Delta}^{\alpha}N^{u(i)}_{\bar{t}}$, ${\Delta}^{\alpha}N^{\ell(i)}_{\bar{t}}$	P[3,1], P[4,j] 2<j<12, P[4,j] 13<j<12	Moles of isotopic species α contained in the increment returned to the mantle, new (jth) segment of upper crust, and new (jth segment of lower crust, respectively, representing the redistribution from an orogene at time, t.[1]
${}^{\alpha}E^{m}$, ${}^{\alpha}E^{u}$, ${}^{\alpha}E^{\ell}$	E1, E2, E3	Enrichment factors for isotopic species α (equal to the concentration of the species in the incremental additions relative to its concentration in the reservoir) for the mantle, upper crust, and lower crust, respectively.
${}^{\alpha}F^{m}$, ${}^{\alpha}F^{u}$, ${}^{\alpha}F^{\ell}$	F1, F2, F3	Partitioning ratios for isotopic species α (equal to the concentration ratios of the species among the returning components from an orogene) for the mantle, upper crust, and lower crust, respectively. ${}^{\alpha}F^{m} + {}^{\alpha}F^{u} + {}^{\alpha}F^{\ell} = 1$.

Table 1. Algebraic notations and their equivalent BASIC representation of the plumbotectonics model (continued)

Algebraic	BASIC	Descriptions
	Other Program Variables	
	L1, L2, L3	Decay constants for [illegible], ^{235}U, and ^{232}Th, respectively
	XØ, YØ, ZØ	Initial $^{206}Pb/^{204}Pb$, $^{207}Pb/^{204}Pb$, and $^{208}Pb/^{204}Pb$ at 4.0 b.y. ago.

[1]As the model is presented here, the matrix array P[n,i] is used to inventory the isotopic species ^{204}Pb; similar arrays, Q[n,i], R[n,i], S[n,i], T[n,i], and U[n,i], represent ^{206}Pb, ^{207}Pb, ^{208}Pb, ^{232}Th, and ^{238}U, respectively. Likewise, the partitioning ratios are designated as F1, F2, F3 for lead, as F4, F5, F6 for thorium, and as F7, F8, F9 for uranium for the mantle, upper crust, and lower crust, respectively.

We adopt the convention of counting geologic time, t, backward from the present, and of numbering orogenies consecutively form the oldest (first) to the youngest (last). the equivalent numerical values of age, t; orogeny, k; upper crust matrix position of M[n,i], P[n,i], etc., u(i); and lower crust matrix position of M[n,i], P[n,i], etc., (i), are given in the following table.

t, b.y.	4.0	3.6	3.2	2.8	2.4	2.0	1.6	1.2	0.8	0.4	0.0
k	1	2	3	4	5	6	7	8	9	10	11
u(i)	2	3	4	5	6	7	8	9	10	11	12
(i)	13	14	15	16	17	18	19	20	21	22	23

PROGRAM EXPLANATION

Each version (I and II) of the main plumbotectonics program, including its five output options (1. = Reservoir mass and isotope content, 2. = Isotopic ratios, 3. = T,Max, 4. = T, Min, and 5. = Reservoir concentrations), is stored as a self-contained file on a tape cartridge, which will operate independently of any auxiliary files. In this form one needs only to activate the program by issuing the RUN command, and then choosing the desired output option. The corresponding version I and II subroutine programs for calculating mean age of isotopes in the crust require substantial modification of the main program, and are themselves also given the status of self-contained files. more general form of the program can easily be created whereby frequently changed numerical values of certain model parameters are stored as separate data files or supplied as input when the program is run.

A brief synopsis of the main program designating lines and functions is provided by the following table.

LINES	FUNCTION
10-70, 290-400, 550-780	Initialize conditions and dimension arrays.
80-280, 410-540	Numerical evaluation of model parameters.
790-2020	Functions for calculating mass and isotope contents of reservoirs.
2030-2190	Output options and directory.
2200-2590	Subroutine for reservoir mass and isotope content.
2600-3250	Subroutine for isotopic ratios.

3260-3420	Subroutine for maximum mean age of crust, T,Max.
3430-3770	Subroutine for minimum mean age of crust, T,Min.
3780-4040	Subroutine for isotope concentrations of reservoirs.

An outline of the plumbotectonics model with the various parameters assigned their appropriate numerical values is present for version I in Table 2 and for version II in Table 3. From this description, all the BASIC variables for each version is listed in Table 4 can be evaluated, and the program can be run straightforwardly with the only remaining decision being that of the output option desired. Some selected results directly obtained from the model (or slight modifications thereof, such as the calculation of average element concentrations in parts per million are given in appendices A and B following the program listings. The reader is referred to the companion paper of Zartman and Doe (in press) for a discussion of the implications of these results to terrestrial lead isotopic patterns.

REFERENCES

Doe, B. R., and Zartman, R. E., 1979, Chapter 2. Plumbotectonics I, The Phanerozoic, in Geochemistry of Hydrothermal Ore Deposits, 2nd Edition (H. L. Barnes, ed.), New York, Wiley-Interscience, p. 22-70.

Zartman, R. E., and Doe, B. R., in press, Plumbotectonics--The model: Tectonophysics.

Table 2. Outline of the plumbotectonics lead-isotope evolution model (version I)

I. Starting conditions (all material in undifferentiated upper mantle to 500 km depth [800×10^{24} g] at 4.0 b.y. ago)

A. Lead isotopic composition: $^{206}Pb/^{204}Pb = 10.36$; $^{207}Pb/^{204}Pb = 12.12$; $^{208}Pb/^{204}Pb = 30.55$

B. Element abundances: $^{204}Pb = 38.0 \times 10^{15}$ moles; $^{238}U = 349 \times 10^{15}$ moles; $^{232}Th = 1335 \times 10^{15}$ moles.

II. Eleven orogenies evenly spaced in time; i.e., at 4.0 b.y., 3.6 b.y., 3.2 b.y., . . ., 0.0 b.y. During each orogeny,

A. Of the material entering the orogene, (1) the mantle contributes 1/8 of itself, at 4.0 b.y., 1/16 at 3.6 b.y., 1/32 at 3.2 b.y., 1/64 at 2.8 b.y., and 1/128 thereafter; (2) the upper crust contributes 3/10 of remaining amounts of each older segment (erosional function); and (3) the total crust contributes 1/10 of remaining amounts of each older segment (areal function). Components (2) and (3) do not pertain to the first orogeny.

B. The contents of the orogene are chemically and isotopically homogenized.

C. 2.6×10^{24} g of new upper crust and 2.6×10^{24} g of new lower crust are created, and remaining material returns from the orogene to the mantle. Lead, uranium, and thorium distribute themselves according to the following partitioning ratios.

Element	Mantle	Upper crust	Lower crust
Lead	0.028	0.754	0.218
Uranium	0.024	0.854	0.122
Thorium	0.020	0.788	0.192

Table 3. Outline of the plumbotectonics lead-isotope evolution model (version II)

I. Starting conditions (all material in undifferentiated upper mantle to 500 km depth [800×10^{24} g] at 4.0 b.y. ago

A. Lead isotopic composition: $^{206}Pb/^{204}Pb = 10.32$; $^{207}Pb/^{204}Pb = 12.12$; $^{208}Pb/^{204}Pb = 30.56$.

B. Element abundances: $^{204}Pb = 37.0 \times 10^{15}$ moles; $^{238}U = 344 \times 10^{15}$ moles; $^{232}Th = 1310 \times 10^{15}$ moles.

II. Eleven orogenies evenly spaced in time; i.e., at 4.0 b.y., 3.6 b.y., 3.2 b.y., . . ., 0.0 b.y. During each orogeny,

A. Of the material entering the orogene, (1) the mantle contributes 1/64 of itself, at 4.0 b.y., 1/16 at 3.2 b.y.and 2.8 b.y., 1/32 at 2.4 b.y., 2.0 b.y., and 1.6 b.y., and 1/64 thereafter; (2) the upper crust contributes 3/10 of remaining amounts of each older segment (erosional function); and (3) the total crust contributes 1/10 of remaining amounts of each older segment (areal function). Components (2) and (3) do not pertain to the first orogeny,

B. The contents of the orogene are chemically and isotopically homogenized.

C. 2.6×10^{24} g of new upper crust and 2.6×10^{24} g of new lower crust are created, and remaining material returns from the orogene to the mantle. Lead, uranium, and thorium distribute themselves according to the following partitioning ratios.

Element	Mantle	Upper crust	Lower crust
Lead	0.036	0.742	0.222
Uranium	0.024	0.852	0.124
Thorium	0.020	0.786	0.194

Table 4. Numerical evaluation of the plumbotectonics model parameters

Description	BASIC Representation		Numerical Values		
	Line	Symbol	Version I	Version II	Units
^{238}U decay constant..........	8Ø	L1	0.155125	0.155125	$\times 10^{-9}$ yr^{-1}
^{235}U decay constant..........	9Ø	L2	0.98485	0.98485	$\times 10^{-9}$ yr^{-1}
^{232}Th decay constant.........	1ØØ	L3	0.049475	0.049475	$\times 10^{-9}$ yr^{-1}
Initial $^{206}Pb/^{204}Pb$..........	11Ø	XØ	10.36	10.32	
Initial $^{207}Pb/^{204}Pb$..........	12Ø	YØ	12.12	12.12	
Initial $^{208}Pb/^{204}Pb$..........	13Ø	ZØ	30.55	30.56	
Mantle enrichment factor......	14Ø	E1	4	4	
Upper crust enrichment factor.....................	15Ø	E2	1	1	
Lower crust enrichment factor.....................	16Ø	E3	1	1	
Mantle partitioning ratio (Pb)......................	17Ø	F1	0.028	0.036	
Upper crust partitioning ratio (Pb).................	18Ø	F2	0.754	0.742	
Lower crust partitioning ratio (Pb).................	19Ø	F3	0.218	0.222	
Mantle partitioning ratio (Th)	2ØØ	F4	0.020	0.020	

Table 4. Numerical evaluation of the plumbotectonics model parameters (Cont'd)

Description	BASIC Representation		Numerical Values		
	Line	Symbol	Version I	Version II	Units
Upper crust partitioning ratio (Th)................	21∅	F5	0.788	0.786	
Lower crust partitioning ratio (Th)................	22∅	F6	0.192	0.194	
Mantle partitioning ratio (U)	23∅	F7	0.024	0.024	
Upper crust partitioning ratio (U).................	24∅	F8	0.854	0.852	
Lower crust partitioning ratio (U).................	25∅	F9	0.122	0.124	
Mass of new segment of upper crust................	26∅	U	2.6	2.6	$x10^{24}$ grams
Mass of a new segment of lower crust................	27∅	L	2.6	2.6	$x10^{24}$ grams
Time of first orogeny........	28∅	T	4	4	$x10^{9}$ years
Initial mass of mantle.......	41∅	M[1,1]	800	800	$x10^{24}$ grams
Initial moles of ^{204}Pb in the mantle................	42∅	P[1,1]	38	37	$x10^{15}$ moles
Initial moles of ^{232}Th in the mantle....................	46∅	T[1,1]	1335	1310	$x10^{15}$ moles
Initial moles of ^{238}U in the mantle....................	47∅	U[1,1]	349	344	$x10^{15}$ moles

Table 4. Numerical evaluation of the plumbotectonics model parameters (Cont'd)

Description	BASIC Representation		Numerical Values		
	Line	Symbol	Version I	Version II	Units
Mass fraction of mantle contributing to the creation of an orogene at time t.....	500-510	A[k]	1/8	1/64	at 4.0 b.y.
			1/16	1/32	at 3.6 b.y.
			1/32	1/16	at 3.2 b.y.
			1/64	1/16	at 2.8 b.y
			1/128	1/32	at 2.4 b.y
			1/128	1/32	at 2.0 b.y.
			1/128	1/32	at 1.6 b.y.
			1/128	1/64	at 1.2 b.y.
			1/128	1/64	at 0.8 b.y.
			1/128	1/64	at 0.4 b.y.
			1/128	1/64	at 0.0 b.y.
Erosional function, f_v........	520	B[k]	0.3	0.3	all times
Areal function, f_h............	530	C[k]	0.1	0.1	all times

APPENDIX A: PLUMBOTECTONICS MODEL (VERSION I)

1. LISTING OF MAIN PROGRAM

```
10 REM PLUMBOTECTONICS MODEL OF ZARTMAN & DOE (1980)  -  VERSION I
20 REM U.S.G.S. OPEN-FILE REPORT 80-1088
30 G=0
40 D1=0
50 D2=0
60 DIM MS[4,23],PS[4,23],QS[4,23],RS[4,23],SS[4,23],TS[4,23],US[4,23]
70 DIM AS[11],BS[11],CS[11]
80 L1=0.155125
90 L2=0.98485
100 L3=0.049475
110 X0=10.36
120 Y0=12.12
130 Z0=30.55
140 E1=4
150 E2=1
160 E3=1
170 F1=0.028
180 F2=0.754
190 F3=0.218
200 F4=0.02
210 F5=0.788
220 F6=0.192
230 F7=0.024
240 F8=0.854
250 F9=0.122
260 U=2.6
270 L=2.6
280 T=4
290 FIXED 4
300 FOR I=1 TO 4
310 FOR J=1 TO 23
320 M[I,J]=0
330 P[I,J]=0
340 Q[I,J]=0
350 R[I,J]=0
360 S[I,J]=0
370 T[I,J]=0
380 U[I,J]=0
390 NEXT J
400 NEXT I
410 M[1,1]=800
420 P[1,1]=38
430 Q[1,1]=P[1,1]*X0
440 R[1,1]=P[1,1]*Y0
450 S[1,1]=P[1,1]*Z0
460 T[1,1]=1335
470 U[1,1]=349
480 FOR I=1 TO 11
490 READ A[I]
500 DATA 0.125,0.0625,0.03125,0.015625,0.0078125,0.0078125
```

.
.

```
510 DATA 0.0078125,0.0078125,0.0078125,0.0078125,0.0078125
520 B[I]=0.3
530 C[I]=0.1
540 NEXT I
550 K=1
560 N=2
570 M=13
580 M0=0
590 M1=0
600 M2=0
610 P0=0
620 P1=0
630 P2=0
640 Q0=0
650 Q1=0
660 Q2=0
670 R0=0
680 R1=0
690 R2=0
700 S0=0
710 S1=0
720 S2=0
730 T0=0
740 T1=0
750 T2=0
760 U0=0
770 U1=0
780 U2=0
790 M[2,1]=M[1,1]*A[K]
800 FOR J=2 TO N
810 M[2,J]=M[1,J]*(B[J-1]+C[J-1]-B[J-1]*C[J-1])
820 M1=M[2,J]+M1
830 NEXT J
840 FOR J=13 TO M
850 M[2,J]=M[1,J]*C[J-12]
860 M2=M[2,J]+M2
870 NEXT J
880 M0=M[2,1]+M1+M2
890 M[3,1]=M0-(U+L)
900 M[4,1]=M[1,1]-M[2,1]+M[3,1]
910 FOR J=2 TO N
920 M[4,J]=M[1,J]-M[2,J]
930 NEXT J
940 FOR J=13 TO M
950 M[4,J]=M[1,J]-M[2,J]
960 NEXT J
970 M[4,N]=U
980 M[4,M]=L
990 IF G=3 THEN 3290
1000 IF G=4 THEN 3460
1010 P[2,1]=P[1,1]*A[K]*E1
1020 FOR J=2 TO N
1030 P[2,J]=P[1,J]*(B[J-1]+C[J-1]-B[J-1]*C[J-1])*E2
1040 P1=P[2,J]+P1
1050 P[4,J]=P[1,J]-P[2,J]
1060 NEXT J
```

```
1070 FOR J=13 TO M
1080 P[2,J]=P[1,J]*C[J-12]*E3
1090 P2=P[2,J]+P2
1100 P[4,J]=P[1,J]-P[2,J]
1110 NEXT J
1120 P0=P[2,1]+P1+P2
1130 F=M[3,1]*F1+M[4,N]*F2+M[4,M]*F3
1140 P[3,1]=P0*(M[3,1]*F1/F)
1150 P[4,1]=P[1,1]-P[2,1]+P[3,1]
1160 P[4,N]=P0*(M[4,N]*F2/F)
1170 P[4,M]=P0*(M[4,M]*F3/F)
1180 Q[2,1]=Q[1,1]*A[K]*E1
1190 FOR J=2 TO N
1200 Q[2,J]=Q[1,J]*(B[J-1]+C[J-1]-B[J-1]*C[J-1])*E2
1210 Q1=Q[2,J]+Q1
1220 Q[4,J]=Q[1,J]-Q[2,J]
1230 NEXT J
1240 FOR J=13 TO M
1250 Q[2,J]=Q[1,J]*C[J-12]*E3
1260 Q2=Q[2,J]+Q2
1270 Q[4,J]=Q[1,J]-Q[2,J]
1280 NEXT J
1290 Q0=Q[2,1]+Q1+Q2
1300 Q[3,1]=Q0*(M[3,1]*F1/F)
1310 Q[4,1]=Q[1,1]-Q[2,1]+Q[3,1]
1320 Q[4,N]=Q0*(M[4,N]*F2/F)
1330 Q[4,M]=Q0*(M[4,M]*F3/F)
1340 R[2,1]=R[1,1]*A[K]*E1
1350 FOR J=2 TO N
1360 R[2,J]=R[1,J]*(B[J-1]+C[J-1]-B[J-1]*C[J-1])*E2
1370 R1=R[2,J]+R1
1380 R[4,J]=R[1,J]-R[2,J]
1390 NEXT J
1400 FOR J=13 TO M
1410 R[2,J]=R[1,J]*C[J-12]*E3
1420 R2=R[2,J]+R2
1430 R[4,J]=R[1,J]-R[2,J]
1440 NEXT J
1450 R0=R[2,1]+R1+R2
1460 R[3,1]=R0*(M[3,1]*F1/F)
1470 R[4,1]=R[1,1]-R[2,1]+R[3,1]
1480 R[4,N]=R0*(M[4,N]*F2/F)
1490 R[4,M]=R0*(M[4,M]*F3/F)
1500 S[2,1]=S[1,1]*A[K]*E1
1510 FOR J=2 TO N
1520 S[2,J]=S[1,J]*(B[J-1]+C[J-1]-B[J-1]*C[J-1])*E2
1530 S1=S[2,J]+S1
1540 S[4,J]=S[1,J]-S[2,J]
1550 NEXT J
1560 FOR J=13 TO M
1570 S[2,J]=S[1,J]*C[J-12]*E3
1580 S2=S[2,J]+S2
1590 S[4,J]=S[1,J]-S[2,J]
1600 NEXT J
1610 S0=S[2,1]+S1+S2
1620 S[3,1]=S0*(M[3,1]*F1/F)
```

```
1630 S[4,1]=S[1,1]-S[2,1]+S[3,1]
1640 S[4,N]=S0*(M[4,N]*F2/F)
1650 S[4,M]=S0*(M[4,M]*F3/F)
1660 T[2,1]=T[1,1]*A[K]*E1
1670 FOR J=2 TO N
1680 T[2,J]=T[1,J]*(B[J-1]+C[J-1]-B[J-1]*C[J-1])*E2
1690 T1=T[2,J]+T1
1700 T[4,J]=T[1,J]-T[2,J]
1710 NEXT J
1720 FOR J=13 TO M
1730 T[2,J]=T[1,J]*C[J-12]*E3
1740 T2=T[2,J]+T2
1750 T[4,J]=T[1,J]-T[2,J]
1760 NEXT J
1770 T0=T[2,1]+T1+T2
1780 F=M[3,1]*F4+M[4,N]*F5+M[4,M]*F6
1790 T[3,1]=T0*(M[3,1]*F4/F)
1800 T[4,1]=T[1,1]-T[2,1]+T[3,1]
1810 T[4,N]=T0*(M[4,N]*F5/F)
1820 T[4,M]=T0*(M[4,M]*F6/F)
1830 U[2,1]=U[1,1]*A[K]*E1
1840 FOR J=2 TO N
1850 U[2,J]=U[1,J]*(B[J-1]+C[J-1]-B[J-1]*C[J-1])*E2
1860 U1=U[2,J]+U1
1870 U[4,J]=U[1,J]-U[2,J]
1880 NEXT J
1890 FOR J=13 TO M
1900 U[2,J]=U[1,J]*C[J-12]*E3
1910 U2=U[2,J]+U2
1920 U[4,J]=U[1,J]-U[2,J]
1930 NEXT J
1940 U0=U[2,1]+U1+U2
1950 F=M[3,1]*F7+M[4,N]*F8+M[4,M]*F9
1960 U[3,1]=U0*(M[3,1]*F7/F)
1970 U[4,1]=U[1,1]-U[2,1]+U[3,1]
1980 U[4,N]=U0*(M[4,N]*F8/F)
1990 U[4,M]=U0*(M[4,M]*F9/F)
2000 IF G=1 THEN 2240
2010 IF G=2 THEN 2640
2020 IF G=5 THEN 3850
2030 DISP "OUTPUT OPTION? ENTER 9 FOR LIST";
2040 INPUT G
2050 IF G#9 THEN 2150
2060 PRINT "      * * * * LIST OF OUTPUT OPTIONS * * * *"
2070 PRINT
2080 PRINT "      1  RESERVOIR MASS AND ISOTOPE CONTENT"
2090 PRINT "      2  ISOTOPIC RATIOS"
2100 PRINT "      3  T,MAX"
2110 PRINT "      4  T,MIN"
2120 PRINT "      5  RESERVOIR CONCENTRATIONS"
2130 PRINT
2140 GOTO 2030
2150 IF G=1 THEN 2210
2160 IF G=2 THEN 2610
2170 IF G=3 THEN 3270
2180 IF G=4 THEN 3440
```

```
2190 IF G=5 THEN 3790
2200 REM SUB-ROUTINE FOR RESERVOIR MASS AND ISOTOPE CONTENT
2210 PRINT "           * * * * RESERVOIR MASS AND ISOTOPIC CONTENT * * * *"
2220 PRINT
2230 PRINT
2240 PRINT "T="T
2250 PRINT
2260 PRINT "  I =","    1","    2","    3","    4"
2270 PRINT
2280 FOR J=1 TO 23
2290 FORMAT "M(I,",F3.0,")",F15.4,F15.4,F15.4,F15.4
2300 WRITE (15,2290)J,M[1,J],M[2,J],M[3,J],M[4,J]
2310 FORMAT "P(I,",F3.0,")",F15.4,F15.4,F15.4,F15.4
2320 WRITE (15,2310)J,P[1,J],P[2,J],P[3,J],P[4,J]
2330 FORMAT "Q(I,",F3.0,")",F15.4,F15.4,F15.4,F15.4
2340 WRITE (15,2330)J,Q[1,J],Q[2,J],Q[3,J],Q[4,J]
2350 FORMAT "R(I,",F3.0,")",F15.4,F15.4,F15.4,F15.4
2360 WRITE (15,2350)J,R[1,J],R[2,J],R[3,J],R[4,J]
2370 FORMAT "S(I,",F3.0,")",F15.4,F15.4,F15.4,F15.4
2380 WRITE (15,2370)J,S[1,J],S[2,J],S[3,J],S[4,J]
2390 FORMAT "T(I,",F3.0,")",F15.4,F15.4,F15.4,F15.4
2400 WRITE (15,2390)J,T[1,J],T[2,J],T[3,J],T[4,J]
2410 FORMAT "U(I,",F3.0,")",F15.4,F15.4,F15.4,F15.4
2420 WRITE (15,2410)J,U[1,J],U[2,J],U[3,J],U[4,J]
2430 PRINT
2440 NEXT J
2450 FOR J=1 TO 23
2460 M[1,J]=M[4,J]
2470 P[1,J]=P[4,J]
2480 T[1,J]=T[4,J]
2490 U[1,J]=U[4,J]
2500 Q[1,J]=Q[4,J]+U[4,J]*(EXP(L1*T)-EXP(L1*(T-0.4)))
2510 R[1,J]=R[4,J]+(U[4,J]/137.88)*(EXP(L2*T)-EXP(L2*(T-0.4)))
2520 S[1,J]=S[4,J]+T[4,J]*(EXP(L3*T)-EXP(L3*(T-0.4)))
2530 NEXT J
2540 T=T-0.4
2550 K=K+1
2560 N=N+1
2570 M=M+1
2580 IF N=13 THEN 4040
2590 GOTO 580
2600 REM SUB-ROUTINE FOR ISOTOPIC RATIOS
2610 PRINT "                         * * * * ISOTOPIC RATIOS * * * *"
2620 PRINT
2630 PRINT "   X   ","   Y   ","   Z   ","  MU   "," KAPPA "
2640 PRINT
2650 PRINT "T="T
2660 PRINT
2670 PRINT "MANTLE"
2680 IF P[1,1]=0 THEN 2720
2690 PRINT Q[1,1]/P[1,1],R[1,1]/P[1,1],S[1,1]/P[1,1],U[1,1]/P[1,1],
2700 PRINT T[1,1]/U[1,1]
2710 GOTO 2730
2720 PRINT P[1,1],P[1,1],P[1,1],P[1,1],P[1,1]
2730 PRINT
2740 PRINT "UPPER CRUST"
```

```
2750 FOR J=2 TO 12
2760 IF P[1,J]=0 THEN 2800
2770 PRINT Q[1,J]/P[1,J],R[1,J]/P[1,J],S[1,J]/P[1,J],U[1,J]/P[1,J],
2780 PRINT T[1,J]/U[1,J]
2790 GOTO 2810
2800 PRINT P[1,J],P[1,J],P[1,J],P[1,J],P[1,J]
2810 NEXT J
2820 PRINT
2830 PRINT "LOWER CRUST"
2840 FOR J=13 TO 23
2850 IF P[1,J]=0 THEN 2890
2860 PRINT Q[1,J]/P[1,J],R[1,J]/P[1,J],S[1,J]/P[1,J],U[1,J]/P[1,J],
2870 PRINT T[1,J]/U[1,J]
2880 GOTO 2900
2890 PRINT P[1,J],P[1,J],P[1,J],P[1,J],P[1,J]
2900 NEXT J
2910 PRINT
2920 PRINT "OROGENE"
2930 IF P0=0 THEN 2960
2940 PRINT Q0/P0,R0/P0,S0/P0,U0/P0,T0/U0
2950 GOTO 2970
2960 PRINT P0,P0,P0,P0,P0
2970 PRINT
2980 PRINT "UPPER CRUST CONTRIBUTED TO OROGENE"
2990 IF P1=0 THEN 3020
3000 PRINT Q1/P1,R1/P1,S1/P1,U1/P1,T1/U1
3010 GOTO 3030
3020 PRINT P1,P1,P1,P1,P1
3030 PRINT
3040 PRINT "LOWER CRUST CONTRIBUTED TO OROGENE"
3050 IF P2=0 THEN 3090
3060 PRINT Q2/P2,R2/P2,S2/P2,U2/P2,T2/U2
3070 PRINT
3080 GOTO 3100
3090 PRINT P2,P2,P2,P2,P2
3100 PRINT
3110 FOR J=1 TO 23
3120 M[1,J]=M[4,J]
3130 P[1,J]=P[4,J]
3140 T[1,J]=T[4,J]
3150 U[1,J]=U[4,J]
3160 Q[1,J]=Q[4,J]+U[4,J]*(EXP(L1*T)-EXP(L1*(T-0.4)))
3170 R[1,J]=R[4,J]+(U[4,J]/137.88)*(EXP(L2*T)-EXP(L2*(T-0.4)))
3180 S[1,J]=S[4,J]+T[4,J]*(EXP(L3*T)-EXP(L3*(T-0.4)))
3190 NEXT J
3200 T=T-0.4
3210 K=K+1
3220 N=N+1
3230 M=M+1
3240 IF N=13 THEN 4040
3250 GOTO 580
3260 REM SUB-ROUTINE FOR T,MAX
3270 PRINT "                    * * * * T,MAX * * * *"
3280 PRINT
3290 D1=(M[1,1]-M[4,1])*T+D1
3300 D2=M[1,1]-M[4,1]+D2
```

```
3310 D=D1/D2
3320 IF N<12 THEN 3340
3330 PRINT "AVERAGE AGE OF TOTAL CRUST; T,MAX ="D
3340 FOR J=1 TO 23
3350 M[1,J]=M[4,J]
3360 NEXT J
3370 T=T-0.4
3380 K=K+1
3390 N=N+1
3400 M=M+1
3410 IF N=13 THEN 4040
3420 GOTO 580
3430 REM SUB-ROUTINE FOR T,MIN
3440 PRINT "                    * * * * T,MIN * * * *"
3450 PRINT
3460 FOR J=1 TO 23
3470 M[1,J]=M[4,J]
3480 NEXT J
3490 T=T-0.4
3500 K=K+1
3510 N=N+1
3520 M=M+1
3530 IF N=13 THEN 3550
3540 GOTO 580
3550 I=2
3560 J=13
3570 A1=0
3580 A2=0
3590 B1=0
3600 B2=0
3610 FOR T=4 TO 0 STEP -0.4
3620 A1=M[4,I]*T+A1
3630 A2=M[4,J]*T+A2
3640 B1=M[4,I]+B1
3650 B2=M[4,J]+B2
3660 I=I+1
3670 J=J+1
3680 NEXT T
3690 A=(A1+A2)/(B1+B2)
3700 B=A1/B1
3710 C=A2/B2
3720 PRINT "AVERAGE AGE OF TOTAL CRUST; T,MIN(T.C.)="A
3730 PRINT
3740 PRINT "AVERAGE AGE OF UPPER CRUST; T,MIN(U.C.)="B
3750 PRINT
3760 PRINT "AVERAGE AGE OF LOWER CRUST; T,MIN(L.C.)="C
3770 IF N=13 THEN 4040
3780 REM SUB-ROUTINE FOR RESERVOIR CONCENTRATIONS
3790 PRINT "              * * * * RESERVOIR CONCENTRATIONS * * * *"
3800 PRINT
3810 PRINT "          (EXPRESSED IN UNITS OF 10EXP-09 MOLES/GRAM)"
3820 PRINT
3830 PRINT "   J    "," 204PB "," 238U  "," 232TH "
3840 PRINT
3850 FOR J=1 TO 23
3860 M[1,J]=M[4,J]
```

```
3870 P[1,J]=P[4,J]
3880 T[1,J]=T[4,J]
3890 U[1,J]=U[4,J]
3900 Q[1,J]=Q[4,J]+U[4,J]*(EXP(L1*T)-EXP(L1*(T-0.4)))
3910 R[1,J]=R[4,J]+(U[4,J]/137.88)*(EXP(L2*T)-EXP(L2*(T-0.4)))
3920 S[1,J]=S[4,J]+T[4,J]*(EXP(L3*T)-EXP(L3*(T-0.4)))
3930 NEXT J
3940 T=T-0.4
3950 K=K+1
3960 N=N+1
3970 M=M+1
3980 IF N=13 THEN 4000
3990 GOTO 580
4000 FOR J=1 TO 23
4010 PRINT J,P[4,J]/M[4,J],U[4,J]/M[4,J],T[4,J]/M[4,J]
4020 PRINT
4030 NEXT J
4040 END
```

APPENDIX A: PLUMBOTECTONICS MODEL (VERSION I)

2. LISTING OF SUBROUTINE FOR MEAN AGE OF ISOTOPES IN CRUST

```
10 REM PLUMBOTECTONICS MODEL OF ZARTMAN & DOE (1980)  -  VERSION I
20 REM U.S.G.S. OPEN-FILE REPORT 80-1088
30 DIM MS[4,23],PS[7,23],TS[7,23],US[7,23]
40 DIM AS[11],BS[11],CS[11],DS[3,11]
50 FOR I=1 TO 11
60 READ A[I]
70 DATA 0.125,0.0625,0.03125,0.015625,0.0078125,0.0078125
80 DATA 0.0078125,0.0078125,0.0078125,0.0078125,0.0078125
90 B[I]=0.3
100 C[I]=0.1
110 NEXT I
120 E1=4
130 E2=1
140 E3=1
150 F1=0.028
160 F2=0.754
170 F3=0.218
180 F4=0.02
190 F5=0.788
200 F6=0.192
210 F7=0.024
220 F8=0.854
230 F9=0.122
240 U=2.6
250 L=2.6
260 FIXED 4
270 FOR Z=2 TO 12
280 FOR I=1 TO 4
290 FOR J=1 TO 23
300 M[I,J]=0
310 NEXT J
320 NEXT I
330 FOR I=1 TO 7
340 FOR J=1 TO 23
350 P[I,J]=0
360 T[I,J]=0
370 U[I,J]=0
380 NEXT J
390 NEXT I
400 M[1,1]=800
410 P[1,1]=38
420 T[1,1]=1335
430 U[1,1]=349
440 K=1
450 N=2
460 M=13
470 T=4
480 M0=0
490 M1=0
500 M2=0
```

```
510 P0=0
520 P1=0
530 P2=0
540 P3=0
550 P4=0
560 P5=0
570 T0=0
580 T1=0
590 T2=0
600 T3=0
610 T4=0
620 T5=0
630 U0=0
640 U1=0
650 U2=0
660 U3=0
670 U4=0
680 U5=0
690 M[2,1]=M[1,1]*A[K]
700 FOR J=2 TO N
710 M[2,J]=M[1,J]*(B[J-1]+C[J-1]-B[J-1]*C[J-1])
720 M1=M[2,J]+M1
730 NEXT J
740 FOR J=13 TO M
750 M[2,J]=M[1,J]*C[J-12]
760 M2=M[2,J]+M2
770 NEXT J
780 M0=M[2,1]+M1+M2
790 M[3,1]=M0-(U+L)
800 M[4,1]=M[1,1]-M[2,1]+M[3,1]
810 FOR J=2 TO N
820 M[4,J]=M[1,J]-M[2,J]
830 NEXT J
840 FOR J=13 TO M
850 M[4,J]=M[1,J]-M[2,J]
860 NEXT J
870 M[4,N]=U
880 M[4,M]=L
890 P[2,1]=P[1,1]*A[K]*E1
900 FOR J=2 TO N
910 P[2,J]=P[1,J]*(B[J-1]+C[J-1]-B[J-1]*C[J-1])*E2
920 P1=P[2,J]+P1
930 P[4,J]=P[1,J]-P[2,J]
940 NEXT J
950 FOR J=13 TO M
960 P[2,J]=P[1,J]*C[J-12]*E3
970 P2=P[2,J]+P2
980 P[4,J]=P[1,J]-P[2,J]
990 NEXT J
1000 P0=P[2,1]+P1+P2
1010 F=M[3,1]*F1+M[4,N]*F2+M[4,M]*F3
1020 P[3,1]=P0*(M[3,1]*F1/F)
1030 P[4,1]=P[1,1]-P[2,1]+P[3,1]
1040 P[4,N]=P0*(M[4,N]*F2/F)
1050 P[4,M]=P0*(M[4,M]*F3/F)
1060 IF N=Z THEN 1090
```

```
1070 IF N>Z THEN 1120
1080 GOTO 1250
1090 P[7,N]=P[2,1]*(M[4,N]*F2/F)
1100 P[7,M]=P[2,1]*(M[4,M]*F3/F)
1110 GOTO 1250
1120 FOR J=2 TO N
1130 P[6,J]=P[5,J]*(B[J-1]+C[J-1]-B[J-1]*C[J-1])*E2
1140 P4=P[6,J]+P4
1150 P[7,J]=P[5,J]-P[6,J]
1160 NEXT J
1170 FOR J=13 TO M
1180 P[6,J]=P[5,J]*C[J-12]*E3
1190 P5=P[6,J]+P5
1200 P[7,J]=P[5,J]-P[6,J]
1210 NEXT J
1220 P3=P4+P5
1230 P[7,N]=P3*(M[4,N]*F2/F)
1240 P[7,M]=P3*(M[4,M]*F3/F)
1250 T[2,1]=T[1,1]*A[K]*E1
1260 FOR J=2 TO N
1270 T[2,J]=T[1,J]*(B[J-1]+C[J-1]-B[J-1]*C[J-1])*E2
1280 T1=T[2,J]+T1
1290 T[4,J]=T[1,J]-T[2,J]
1300 NEXT J
1310 FOR J=13 TO M
1320 T[2,J]=T[1,J]*C[J-12]*E3
1330 T2=T[2,J]+T2
1340 T[4,J]=T[1,J]-T[2,J]
1350 NEXT J
1360 T0=T[2,1]+T1+T2
1370 F=M[3,1]*F4+M[4,N]*F5+M[4,M]*F6
1380 T[3,1]=T0*(M[3,1]*F4/F)
1390 T[4,1]=T[1,1]-T[2,1]+T[3,1]
1400 T[4,N]=T0*(M[4,N]*F5/F)
1410 T[4,M]=T0*(M[4,M]*F6/F)
1420 IF N=Z THEN 1450
1430 IF N>Z THEN 1480
1440 GOTO 1610
1450 T[7,N]=T[2,1]*(M[4,N]*F5/F)
1460 T[7,M]=T[2,1]*(M[4,M]*F6/F)
1470 GOTO 1610
1480 FOR J=2 TO N
1490 T[6,J]=T[5,J]*(B[J-1]+C[J-1]-B[J-1]*C[J-1])*E2
1500 T4=T[6,J]+T4
1510 T[7,J]=T[5,J]-T[6,J]
1520 NEXT J
1530 FOR J=13 TO M
1540 T[6,J]=T[5,J]*C[J-12]*E3
1550 T5=T[6,J]+T5
1560 T[7,J]=T[5,J]-T[6,J]
1570 NEXT J
1580 T3=T4+T5
1590 T[7,N]=T3*(M[4,N]*F5/F)
1600 T[7,M]=T3*(M[4,M]*F6/F)
1610 U[2,1]=U[1,1]*A[K]*E1
1620 FOR J=2 TO N
```

```
1630 U[2,J]=U[1,J]*(B[J-1]+C[J-1]-B[J-1]*C[J-1])*E2
1640 U1=U[2,J]+U1
1650 U[4,J]=U[1,J]-U[2,J]
1660 NEXT J
1670 FOR J=13 TO M
1680 U[2,J]=U[1,J]*C[J-12]*E3
1690 U2=U[2,J]+U2
1700 U[4,J]=U[1,J]-U[2,J]
1710 NEXT J
1720 U0=U[2,1]+U1+U2
1730 F=M[3,1]*F7+M[4,N]*F8+M[4,M]*F9
1740 U[3,1]=U0*(M[3,1]*F7/F)
1750 U[4,1]=U[1,1]-U[2,1]+U[3,1]
1760 U[4,N]=U0*(M[4,N]*F8/F)
1770 U[4,M]=U0*(M[4,M]*F9/F)
1780 IF N=Z THEN 1810
1790 IF N>Z THEN 1840
1800 GOTO 1980
1810 U[7,N]=U[2,1]*(M[4,N]*F8/F)
1820 U[7,M]=U[2,1]*(M[4,M]*F9/F)
1830 GOTO 1980
1840 FOR J=2 TO N
1850 U[6,J]=U[5,J]*(B[J-1]+C[J-1]-B[J-1]*C[J-1])*E2
1860 U4=U[6,J]+U4
1870 U[7,J]=U[5,J]-U[6,J]
1880 NEXT J
1890 FOR J=13 TO M
1900 U[6,J]=U[5,J]*C[J-12]*E3
1910 U5=U[6,J]+U5
1920 U[7,J]=U[5,J]-U[6,J]
1930 NEXT J
1940 U3=U4+U5
1950 U[7,N]=U3*(M[4,N]*F8/F)
1960 U[7,M]=U3*(M[4,M]*F9/F)
1970 FOR J=1 TO 23
1980 FOR J=1 TO 23
1990 M[1,J]=M[4,J]
2000 P[1,J]=P[4,J]
2010 P[5,J]=P[7,J]
2020 T[1,J]=T[4,J]
2030 T[5,J]=T[7,J]
2040 U[1,J]=U[4,J]
2045 U[5,J]=U[7,J]
2050 NEXT J
2060 T=T-0.4
2070 K=K+1
2080 N=N+1
2090 M=M+1
2100 IF N=13 THEN 2120
2110 GOTO 480
2120 A=0
2130 B=0
2140 FOR J=Z TO 12
2150 A=P[7,J]+A
2160 B=P[7,J+11]+B
2170 NEXT J
2180 D[1,Z-1]=A+B
```

```
2190 FOR J=Z TO 12
2200 PRINT 4.8-J*0.4,P[7,J],P[7,J+11]
2210 NEXT J
2220 PRINT
2230 PRINT "MOLES OF 204PB OF AGE"4.8-Z*0.4"B.Y. IN U.C. ="A
2240 PRINT
2250 PRINT "MOLES OF 204PB OF AGE"4.8-Z*0.4"B.Y. IN L.C. ="B
2260 PRINT
2270 PRINT
2280 A=0
2290 B=0
2300 FOR J=Z TO 12
2310 A=T[7,J]+A
2320 B=T[7,J+11]+B
2330 NEXT J
2340 D[2,Z-1]=A+B
2350 FOR J=Z TO 12
2360 PRINT 4.8-J*0.4,T[7,J],T[7,J+11]
2370 NEXT J
2380 PRINT
2390 PRINT "MOLES OF 232TH OF AGE"4.8-0.4*Z"B.Y. IN U.C. ="A
2400 PRINT
2410 PRINT "MOLES OF 232TH OF AGE"4.8-0.4*Z"B.Y. IN L.C. ="B
2420 PRINT
2430 PRINT
2440 A=0
2450 B=0
2460 FOR J=Z TO 12
2470 A=U[7,J]+A
2480 B=U[7,J+11]+B
2490 NEXT J
2500 D[3,Z-1]=A+B
2510 FOR J=Z TO 12
2520 PRINT 4.8-J*0.4,U[7,J],U[7,J+11]
2530 NEXT J
2540 PRINT
2550 PRINT "MOLES OF 238U OF AGE"4.8-0.4*Z"B.Y. IN U.C. ="A
2560 PRINT
2570 PRINT "MOLES OF 238U OF AGE"4.8-0.4*Z"B.Y. IN L.C. ="B
2580 PRINT
2590 PRINT
2600 NEXT Z
2610 PRINT
2620 PRINT
2630 PRINT
2640 C1=0
2650 D1=0
2660 C2=0
2670 D2=0
2680 C3=0
2690 D3=0
2700 FOR J=1 TO 11
2710 C1=D[1,J]*(4.4-0.4*J)+C1
2720 D1=D[1,J]+D1
2730 C2=D[2,J]*(4.4-0.4*J)+C2
2740 D2=D[2,J]+D2
```

```
2750 C3=D[3,J]*(4.4-0.4*J)+C3
2760 D3=D[3,J]+D3
2770 NEXT J
2780 PRINT "MEAN AGE OF 204PB ="C1/D1"M.Y."
2790 PRINT
2800 PRINT "MEAN AGE OF 232TH ="C2/D2"M.Y."
2810 PRINT
2820 PRINT "MEAN AGE OF 238U ="C3/D3"B.Y."
2830 PRINT
2840 PRINT
2850 PRINT
2860 PRINT "C1="C1,"C2="C2,"C3="C3
2870 PRINT "D1="D1,"D2="D2,"D3="D3
2880 END
```

Mass distribution for the various groupings of reservoirs as a function of time. $M^{u(1)}_{4.0}$ and $M^{\ell(1)}_{4.0}$ are also given to illustrate the decrease in size of crustal reservoirs with successive orogenies. [All masses in units of 10^{24} g]

i	t (b.y.)	M^{m}_{t}	$\sum M^{u(i)}_{t}$	$\sum M^{\ell(i)}_{t}$	$M^{u(1)}_{4.0}$	$M^{\ell(1)}_{4.0}$
0	4.0	800.0	0.000	0.000	0.000	0.000
1	4.0	794.8	2.600	2.600	2.600	2.600
2	3.6	790.8	4.238	4.940	1.638	2.340
3	3.2	787.7	5.270	7.046	1.032	2.106
4	2.8	785.1	5.920	8.941	.650	1.895
5	2.4	783.0	6.330	10.647	.410	1.706
6	2.0	781.2	6.588	12.183	.258	1.535
7	1.6	779.7	6.750	13.564	.163	1.382
8	1.2	778.3	6.853	14.808	.102	1.244
9	0.8	777.1	6.917	15.927	.065	1.119
10	0.4	776.1	6.958	16.934	.041	1.007
11	0.0	775.2	6.983	17.841	.026	.907

Lead isotopic composition for reservoir growth curves generated by the plumbotectonics model (version I)

t (b.y.)	Mantle			Orogene			Upper Crust			Lower Crust		
	$^{206}Pb/^{204}Pb$	$^{207}Pb/^{204}Pb$	$^{208}Pb/^{204}Pb$	$^{206}Pb/^{204}Pb$	$^{207}Pb/^{204}Pb$	$^{208}Pb/^{204}Pb$	$^{206}Pb/^{204}Pb$	$^{207}Pb/^{204}Pb$	$^{208}Pb/^{204}Pb$	$^{206}Pb/^{204}Pb$	$^{207}Pb/^{204}Pb$	$^{208}Pb/^{204}Pb$
4.0	10.36	12.12	30.55	10.36	12.12	30.55	10.36	12.12	30.55	10.36	12.12	30.55
3.6	11.36	13.21	31.34	11.42	13.27	31.40	11.61	13.48	31.57	10.98	12.79	31.41
3.2	12.30	13.94	32.11	12.47	14.09	32.30	12.69	14.29	32.46	11.76	13.46	32.24
2.8	13.19	14.44	32.86	13.50	14.67	33.20	13.72	14.83	33.35	12.52	13.95	33.06
2.4	14.02	14.78	33.59	14.47	15.06	34.11	14.70	15.20	34.23	13.27	14.33	33.89
2.0	14.80	15.00	34.30	15.35	15.29	34.95	15.64	15.44	35.10	14.00	14.61	34.70
1.6	15.54	15.15	35.00	16.17	15.44	35.77	16.51	15.59	35.94	14.72	14.83	35.51
1.2	16.23	15.26	35.69	16.92	15.53	36.56	17.30	15.67	36.76	15.40	15.00	36.29
0.8	16.88	15.33	36.37	17.62	15.58	37.34	18.04	15.71	37.55	16.06	15.13	37.06
0.4	17.50	15.38	37.03	18.27	15.61	38.09	18.71	15.73	38.32	16.69	15.22	37.82
0.0	18.08	15.42	37.68	18.88	15.63	38.82	19.33	15.73	39.06	17.29	15.30	38.56

Ending conditions (prevailing today at time of final orogeny) for plumbotectonics model (version I)

A. Lead isotopic composition:

Environment	$^{206}Pb/^{204}Pb$	$^{207}Pb/^{204}Pb$	$^{208}Pb/^{204}Pb$	$^{238}Pb/^{204}Pb$	$^{232}Th/^{238}U$
Mantle - - - - - - - - -	18.08	15.42	37.68	8.90	3.58
Orogene - - - - - - - - -	18.88	15.63	38.82	10.88	3.62
Upper crust contributed to orogene - - - - -	19.33	15.73	39.06	12.22	3.40
Lower crust contributed to orogene - - - - -	17.29	15.30	38.56	5.90	5.94

B. Element abundances and concentrations:

Reservoir	Mass	Abundances ($\times 10^{15}$ moles)			Average concentrations (ppm)		
	($\times 10^{24}$g)	^{204}Pb	^{238}U	^{232}Th	Pb	U	Th
Mantle - - - - - -	775.2	19.5	174	619	0.39	0.054	0.19
Upper Crust - - -	7.0	10.3	127	430	23.5	4.4	14.3
Lower crust - - -	17.8	8.2	48	286	6.8	0.64	3.7

NOTE: Modifications in the model equations (see text) and minor calculation errors in the original treatment cause this table to differ slightly from the corresponding table of Doe and Zartman (1979). The results are in close agreement and represent essentially the same version of the plumbotectonics model.

APPENDIX B: PLUMBOTECTONICS MODEL (VERSION II)

1. LISTING OF MAIN PROGRAM (PROGRAM IS SIMILAR TO THAT OF VERSION I GIVEN IN APPENDIX A EXCEPT THAT THE FOLLOWING LINES SHOULD BE SUBSTITUTED FOR THE EQUIVALENT LINES OF VERSION I)

```
10 REM PLUMBOTECTONICS MODEL OF ZARTMAN & DOE (1980)  -  VERSION II
110 X0=10.32
130 Z0=30.56
170 F1=0.036
180 F2=0.742
190 F3=0.222
210 F5=0.786
220 F6=0.194
240 F8=0.852
250 F9=0.124
420 P[1,1]=37
460 T[1,1]=1310
470 U[1,1]=344
500 DATA 0.015625,0.03125,0.0625,0.0625,0.03125,0.03125
510 DATA 0.03125,0.015625,0.015625,0.015625,0.015625
```

2. LISTING OF SUBROUTINE FOR MEAN AGE OF ISOTOPES IN CRUST (PROGRAM IS SIMILAR TO THAT OF VERSION I GIVEN IN APPENDIX A EXCEPT THAT THE FOLLOWING LINES SHOULD BE SUBSTITUTED FOR THE EQUIVALENT LINES OF VERSION I)

```
10 REM PLUMBOTECTONICS MODEL OF ZARTMAN & DOE (1980)  -  VERSION II
80 DATA 0.015625,0.03125,0.0625,0.0625,0.03125,0.03125
90 DATA 0.03125,0.015625,0.015625,0.015625,0.015625
160 F1=0.036
170 F2=0.742
180 F3=0.222
200 F5=0.786
210 F6=0.194
230 F8=0.852
240 F9=0.124
420 P[1,1]=37
430 T[1,1]=1310
440 U[1,1]=344
```

Lead isotopic composition for reservoir growth curves generated by the plumbotectonics model (version II)

t (b.y.)	Mantle			Orogene			Upper Crust			Lower Crust		
	$^{206}Pb/^{204}Pb$	$^{207}Pb/^{204}Pb$	$^{208}Pb/^{204}Pb$	$^{206}Pb/^{204}Pb$	$^{207}Pb/^{204}Pb$	$^{208}Pb/^{204}Pb$	$^{206}Pb/^{204}Pb$	$^{207}Pb/^{204}Pb$	$^{208}Pb/^{204}Pb$	$^{206}Pb/^{204}Pb$	$^{207}Pb/^{204}Pb$	$^{208}Pb/^{204}Pb$
4.0	10.32	12.12	30.56	10.32	12.12	30.56	10.32	12.12	30.56	10.32	12.12	30.56
3.6	11.36	13.25	31.40	11.38	13.26	31.41	11.54	13.44	31.48	10.91	12.76	31.32
3.2	12.32	14.00	32.22	12.36	14.03	32.24	12.62	14.25	32.38	11.81	13.55	32.16
2.8	13.21	14.50	32.99	13.31	14.56	33.07	13.65	14.78	33.30	12.67	14.12	33.02
2.4	14.03	14.83	33.72	14.29	14.96	33.96	14.63	15.13	34.19	13.46	14.51	33.86
2.0	14.80	15.05	34.42	15.17	15.20	34.78	15.59	15.37	35.09	14.21	14.78	34.69
1.6	15.53	15.21	35.11	15.98	15.36	35.58	16.46	15.52	35.94	14.93	14.98	35.51
1.2	16.21	15.32	35.79	16.86	15.49	36.49	17.27	15.60	36.76	15.62	15.13	36.31
0.8	16.85	15.39	36.44	17.59	15.55	37.29	18.06	15.66	37.60	16.29	15.24	37.11
0.4	17.45	15.44	37.08	18.27	15.59	38.06	18.78	15.69	38.41	16.93	15.32	37.90
0.0	18.01	15.47	37.71	18.89	15.61	38.81	19.44	15.70	39.19	17.54	15.39	38.67